



### A USEFUL MICRO-COMPUTER PROGRAM

By W. E. Kilby

The last few years have seen a proliferation of portable and hand-held micro-computers (electronic notebooks). Recently, exploration companies have been making increased use of these machines while in the field. Unfortunately, few programs are widely available to handle geology-specific tasks on these versatile tools.

The program described here, THICK calculates the stratigraphic thickness between two points with similar or differing bedding orientations. The program given here (Fig. 43) is written in Microsoft BASIC and will run with little or no modification on most of the popular micro-computers. The small memory requirements (3K bytes) allow the program to run on most all hand-held computers as well as inexpensive computers. Positional and orientation data for both points are entered as x, y, z coordinates and dip direction and dip angles, respectively. The program contains useful routines of interest in structural geology. Charlesworth and Kilby (1980) contains a detailed explanation of the numerical techniques as well as a practical example. Busk (1929) illustrates the equivalent graphical procedures.




Figure 43. Examples of the signing of the thickness results from the two calculation techniques. Points 1 and 2 are the pitch lines projected onto the plane of profile.

```

1  REM***** THICK *****
2  REM*** W.E.KILBY ***
3  REM*****
100 PI=3.14159/180:PK=1.5708
109 REM ARCCOSINE
110 DEF FNAC(Q)=-ATN(Q/SQR(-Q*Q+1))+PK
119 REM ENTER POSITIONAL AND ORIENTATION DATA
120 INPUT"ENTER X,Y,Z COORDS OF O/C 1";X1,Y1,Z1
130 INPUT"ENTER DIP-DIR. AND DIP OF O/C 1";D1,P1
140 INPUT"ENTER X,Y,Z COORDS OF O/C 2";X2,Y2,Z2
150 INPUT"ENTER DIP-DIR. AND DIP OF O/C 2";D2,P2
160 P=(90-P1)*PI:T=(D1+180)*PI
170 GOSUB 580
180 L1=L:M1=M:N1=N
188 REM TEST IF ORIENTATIONS DIFFERENT, IF SAME CALCULATE THICKNESS
189 REM BY MEAN METHOD.
190 IF (D1=D2)AND(P1=P2)THEN L3=L1:M3=M1:N3=N1:GOTO 260
200 P=(90-P2)*PI:T=(D2+180)*PI
210 GOSUB 580
220 L2=L:M2=M:N2=N
229 REM CALCULATE MEAN ORIENTATION
230 L3=COS((FNAC(L1)+FNAC(L2))/2)
240 M3=COS((FNAC(M1)+FNAC(M2))/2)
250 N3=COS((FNAC(N1)+FNAC(N2))/2)
260 TH=(X2-X1)*L3+(Y2-Y1)*M3+(Z2-Z1)*N3
270 PRINT"THICKNESS BY MEAN ORIENTATION METHOD IS = **";
279 REM CALCULATE DIRECTION COSINES
280 PRINT TH;"**":IF (D1=D2) AND (P1=P2) THEN 600
290 L3=L1:M3=M1:N3=N1:L4=L2:M4=M2:N4=N2
300 GOSUB 540
310 IF DE=0 THEN P=0:T=T+PK:GOSUB 580:LB=L:MB=M:NB=N:GOTO 330
320 LB=LL:MB=MM:NB=NN
330 NT=ATN(LB/MB):NP=ATN(NB/SQR(-NB*NB+1))
340 IF (LB<0)AND(MB<0)THEN 370
350 IF MB>0 THEN NT=NT:GOTO 370
360 NT=PK*4+NT
370 T=NT+PK:P=0:GOSUB 580
380 LH=L:MH=M:NH=N:T=NT:P=PK-NP
390 GOSUB 580
400 LV=L:MV=M:NV=N
409 REM CALCULATE COORDINATES OF POINTS ON PROFILE
410 X3=LH*X1+MH*Y1+NH*Z1:Y3=LV*X1+MV*Y1+NV*Z1
420 X4=LH*X2+MH*Y2+NH*Z2:Y4=LV*X2+MV*Y2+NV*Z2
430 L3=L1:M3=M1:N3=N1:L4=L2:M4=M2:N4=N2
440 GOSUB 540
449 REM CALCULATE PITCH ANGLES
450 AN=LL*LH+MM*MH+NN*NH:A1=FNAC(AN)
460 L3=L2:M3=M2:N3=N2:GOSUB 540
470 AN=LL*LH+MM*MH+NN*NH:A2=FNAC(AN)
480 X5=X3-X4:Y5=Y3-Y4
489 REM CALCULATE BUSK THICKNESS
490 SU=COS(A1)*SIN(A2)-COS(A2)*SIN(A1)
500 PP=SIN(A1)*(Y5*SIN(A2)-X5*COS(A2))/SU
510 QQ=COS(A1)*(Y5*SIN(A2)-X5*COS(A2))/SU
520 TH=SQR(PP^2+QQ^2)-SQR((PP-X5)^2+(QQ-Y5)^2)
530 PRINT"THICKNESS BY BUSK METHOD = ** ";TH;" **":GOTO 600
539 REM CALCULATE ORIENTATION OF INTERSECTION OF TWO PLANES
540 DE=L3*M4-L4*M3:IF DE=0 THEN 570
550 D=(M3*N4-M4*N3)/DE:E=(L4*N3-L3*N4)/DE
560 NN=1/SQR(D^2+E^2+1):LL=D*NN:MM=E*NN
570 RETURN
580 L=SIN(T)*COS(P):M=COS(T)*COS(P):N=-SIN(P)
590 RETURN
600 END

```

Figure 44. A MICROSOFT basic version of the thickness calculation program. Only lines with numbers evenly divisible by 10 need be entered; the rest are comment lines. In its present form the program can be easily adapted to take advantage of machine-specific features.

Two methods of thickness calculation are employed in the program: the mean orientation, and the Busk techniques. If identical orientations are present at both positions this value is used in the mean orientation technique and the Busk method is neglected. When varying orientations are entered, the mean orientation between those entered is used in the mean technique while the two different orientations are used for calculating the Busk thickness. Results from the two techniques are given so the user may compare the results and explain discrepancies, if any. The results are signed, indicating relative positions between data points. In the case of the mean method a negative value indicates that the first data point entered is stratigraphically lower than the second position. In the Busk method a negative value means that the first position entered is further from the centre of curvature of the beds than the second position (Fig. 44). In this program the bedding orientation angles must be relative to the coordinate system used to locate the data-point positions.

The advantage of these numerical techniques over the equivalent graphic procedures are obvious when only one calculation is being undertaken but truly impressive when a large series of values are to be calculated rapidly. A few common uses of this program are:

- (1) Calculating stratigraphic intervals along traverse lines (Charlesworth and Kilby, 1980) or across existing geology maps.
- (2) Calculating stratigraphic thickness in a covered interval during section measuring.
- (3) Predicting horizon depth in boreholes given various bedding orientations or borehole locations.
- (4) Thickness calculations given bedding-to-core angles.
- (5) Thickness calculations given dipmeter values.

As it stands the program calculates one problem at a time; it is easily modified to sum the stratigraphic interval for a series of data points such as outcrops or bedding-to-core angles down a borehole.

Occasionally the user would like more information from the calculations than the stratigraphic thickness. The following values are of potential interest in judging the validity of the calculation or for use in some other task:

- (1) Orientation of the intersection of the two bedding planes, fold-axis orientation.
- (2) The coordinates of the two points projected onto the plane normal to the fold-axis orientation.
- (3) The clockwise pitch angles of the two projected orientations on the plane normal to the fold-axis orientation.

Microsoft BASIC on micro-computers runs interactively. This allows the user to stop the program at any point and determine the value of a variable. The three sets of information described above can be obtained by adding in the following lines to Figure 43:

- 331 PRINT NT,NP — where values are the trend and plunge of the fold-axis orientation in radians.
- 421 PRINT X3, Y3, X4, Y4 — yields the horizontal and vertical coordinates on the plane of the profile for points one and two, respectively.
- 475 PRINT A1,A2 — provides the angles, in radians, of the clockwise pitch of bedding from points one and two on the plane of the profile.

At other times it is desirable to add values calculated outside the program for fold-axis orientations or pitch angles. Again the program can easily be modified to allow introduction of an outside value. Often a standard fold-axis orientation is desired for a large number of calculations within a structural domain. Insertion of the following lines will enable a user-preferred fold-axis orientation to be used:

```
331 INPUT"ENTER NEW TREND AND PLUNGE OF FOLD-AXIS";NT,NP
332 T=NT*PI:P=NP*PI:GOSUB 580:LB=L:MB=M:NB=N
```

Modification of the calculated pitch angle at one or either of the two points is accomplished by first stopping the program and examining the existing values by inserting the following line:

```
475 STOP:PRINT A1,A2
```

The above line yields the present values of pitches in radians; to change these values and start the program execution the following is entered into the computer with the new values replacing the question marks:

```
A1=???:A2=???:CONT
```

The enclosed program (Fig. 43) is a bare-bones version because it does not take advantage of machine-specific commands or mass storage but was designed to be highly portable. It can be typed into any machine with Microsoft BASIC with no changes. Slight modifications may be required for other versions of BASIC. If the thought of typing the whole program is not agreeable it is presently running on a PET and a KAYPRO so any machine compatible with these could get a cassette or diskette copy by sending a blank and a self-addressed mailer to the author. The program could also be easily copied via a telephone hookup. The accompanying Table contains a small set of test data to verify correct entry of the program.

#### TEST DATA

X1	Y1	Z1	DD1	DIP1	X2	Y2	Z2	DD2	DIP2	MEAN	BUSK
0	0	0	30	30	10	10	0	30	30	-6.83	---
0	0	0	30	25	10	10	0	30	38	-7.11	-7.18
0	0	0	30	38	10	10	0	30	25	-7.11	7.18
0	0	0	30	45	10	10	0	150	45	-3.54	-5.66
0	0	0	10	30	10	10	0	70	40	-7.19	5.48

#### REFERENCES

- Busk, H. G. (1929): *Earth Flexures*, Cambridge University Press, London, 106 pp.  
 Charlesworth, H.A.K. and Kilby, W. E. (1981): Calculating Thickness from Outcrop and Drill-Hole Data, *Canadian Petroleum Geology*, Bull., Vol. 29, No. 2, pp. 277-292.